

ROBOT WAR  
INSTRUCTIONS

IT'S YOU AGAINST THE ROBOT. YOU TRY TO GET AS MANY POINTS AS YOU CAN.  
WHEN THE BALLY SAYS "MAX SCORE?" INPUT THE NUMBER YOU WANT AS THE  
SCORE TO END THE GAME AS IN GUNFIGHT.

THE ROBOT IS THE ONE WHO FIRES FIRST AND HAS A FATTER BODY THAN  
THE OTHER FIGURE WHICH IS YOUR MAN. THE ROBOT WILL EITHER FIRE TWO  
LASER SHOTS OR WILL FIRE A PHOTON TORPEDOE THE OTHER WAY.

AFTER THE ROBOT MOVES TO A NEW SPOT, YOU CAN THEN MOVE. TO MOVE,  
MOVE THE JOYSTICK. IT CONTROLS YOUR MAN THE SAME WAY AS IN GUNFIGHT

ONCE YOU GET DONE MOVING YOU CAN FIRE AT THE ROBOT IF YOU'RE  
CLOSE ENOUGH. TO FIRE TO YOUR LEFT TURN THE KNOB ALL THE WAY TO THE LEFT.  
TO FIRE TO YOUR RIGHT TURN THE KNOB ALL THE WAY TO THE RIGHT. THEN AFTER  
YOU TURN THE KNOB THE WAY YOU WANT TO FIRE, PULL THE TRIGGER.

WHEN THE BALLY SAYS "SCORE" THE FIRST NUMBER BELOW IT IS THE ROBOT'S  
SCORE. THE SECOND NUMBER IS YOUR SCORE.

WHEN THE "MAX SCORE" IS REACHED THE BALLY WILL SAY "ROBOT" IF THE  
ROBOT WON AND "YOU" IF YOU WON.

## ROBOT WAR

NOTE: \*indicates multiplication

```

01 CLEAR;INPUT "MAX SCORE?"S
02 &(21)=0;&(20)=0;&(19)=0;P=0;R=0;I=0
05 CLEAR
75 NT=3;A=1
80 GOSUB 120
100 GOSUB 230
105 A=0
110 GOTO 360
120 X=RND(130)-70;Y=RND(60)-30
130 BOX X,Y,6,6,A
140 LINE X-2,Y-3,4;LINE X-2,Y-5,A
150 LINE X+2,Y-3,4;LINE X+2,Y-5,A
160 LINE X-1,Y+3,4;LINE X-1,Y+6,A
170 LINE X+1,Y+3,4;LINE X+1,Y+6,A
180 BOX X,Y+8,7,4,A
190 BOX X,Y+8,2,2,2
200 LINE X+3,Y+1,4;LINE X+6,Y+1,A
210 LINE X+3,Y-1,4;LINE X+10,Y-1,A
220 RETURN
230 L=RND(130)-70;M=RND(60)-30
240 BOX L,M,4,8,A
250 LINE L-2,M-4,4;LINE L-2,M-7,A
260 LINE L+1,M-4,4;LINE L+1,M-7,A
270 LINE L,M+4,4;LINE L,M+6,A
280 BOX L,M+7,4,2,A
290 LINE L-3,M+4,4;LINE L+3,M+4,A
300 LINE L-2,M,4;LINE L-5,M,A
310 LINE L+2,M,4;LINE L+5,M,A
350 RETURN
360 IF L<X GOTO 800
363 Z=X+6;Q=Y+1
365 J=RND(5) 364 FOR V=1 TO 2
370 IF J=1F=5;B=0;GOTO 420
380 IF J=2F=5;B=5;GOTO 420
390 IF J=3F=0;B=5;GOTO 420
400 IF J=4F=-5;B=5;GOTO 420
410 F=-5;B=0
420 &(19)=9;&(20)=22;&(21)=12
423 LINE Z,Q,4;LINE Z+B,Q+F,1
430 IF Z>L-3IF Z<L+3IF Q>M-5IF Q<M+5 GOTO 700
440 Z=Z+B;Q=Q+F
445 IF (Z>74)+(Z<-74)+(Q>38)+(Q<-38)GOTO 460
450 GOTO 423
460 LINE Z,Q,4;LINE Z+B,Q+F,1
465 &(21)=0
470 IF V=1LINE X+6,Y+1,4;LINE Z,Q,2;LINE Z+B,Q+F,2;GOTO 475
473 LINE X+10,Y-1,4;LINE Z,Q,2;LINE Z+B,Q+F,2
475 Z=X+10;Q=Y-1
477 NEXT V
490 A=2;GOSUB 130
500 X=X+RND(30)-15;Y=Y+RND(20)-10
510 A=1;GOSUB 130
520 FOR V=1TO 2
525 A=2;GOSUB 240
530 IF JY(1)=1 M=M+10
540 IF JY(1)=-1 M=M-10
550 IF JX(1)=1 L=L+15

```

```
560 IF JX(1)=-1 L=L-15
570 IF L>70 L=70
571 IF L<-70 L=-70
572 IF M>35 M=35
573 IF M<-35 M=-35
575 A=1; GOSUB 240
580 NEXT V
590 FOR V=1 TO 30
592 IF TR(1)=1 GOTO 600
595 NEXT V
598 GOTO 360
600 IF KN(1)<0 GOTO 645
610 LINE L+6,M,4;LINE L+25,M,1
615 FOR V=1 TO 50
617 &(21)=14;&(19)=20;&(20)=9
618 NEXT V
619 &(21)=0
620 IF L>X-25 IF L<X+15 IF M>Y-4 IF M<Y+3 GOTO 640
630 LINE L+6,M,4;LINE L+25,M,2
637 GOTO 360
640 P=P+1;GOSUB 900
643 GOTO 5
645 LINE L-6,M,4;LINE L-25,M,1
650 FOR V=1 TO 50
660 &(21)=14;&(19)=0;&(20)=9
665 NEXT V
670 &(21)=0
671 LINE L-6,M,4;LINE L-25,M,2
672 IF L>X-25 IF L<X+25 IF M>Y-4 IF M<Y+3 GOTO 640
673 GOTO 360
700 R=R+1; GOSUB 900; GOTO 5
800 Z=X;Q=Y+8
803 &(21)=14;&(19)=65;A=1
810 IF M>Y+15 K=1+RND(4);GOTO 830
820 IF M<Y-15 K=-1-RND(4);GOTO 830
825 K=0-RND(3)
830 BOX Z-8,Q+K,6,2,A
835 IF Z<L IF I=1 I=0;GOTO 490
840 IF Z<L GOTO 860
845 IF Z>L-7 IF Z<L+7 IF Q>M-8 IF Q<M+8 GOTO 870
850 Z=Z-8;Q=Q+K
855 GOTO 830
860 A=2;I=1;&(21)=0;Z=X;Q=Y+8;GOTO 830
870 GOTO 700
900 &(21)=0
905 CY=40;PRINT " SCORE"
910 PRINT #2,R,#6,P
911 IF R=S GOTO 917
912 IF P=S GOTO 918
915 FOR V=1 TO 700; NEXT V
916 RETURN
917 PRINT "ROBOT";STOP
918 PRINT "YOU"
```